
Arquitetura de Computadores

Conjunto de instruções do RAMSES

H Código e
Operando

Endereço
(2º byte)

Instrução Operação N Z C

0x 0000 xxxx - NOP Nenhuma operação
1? 0001 rrmm dd8 STR rr dd8 MEM(dd8) ← rr
2? 0010 rrmm dd8 LDR rr dd8 rr ← MEM(dd8) t t
3? 0011 rrmm dd8 ADD rr dd8 rr ← rr + MEM(dd8) t t t
4? 0100 rrmm dd8 OR rr dd8 rr ← rr∨ MEM(dd8) t t
5? 0101 rrmm dd8 AND rr dd8 rr ← rr∧ MEM(dd8) t t
6? 0110 rrxx - NOT rr rr ← ¬ rr t t
7? 0111 rrmm dd8 SUB rr dd8 rr ← rr - MEM(dd8) t t t ¹
8? 1000 xxmm dd8 JMP dd8 PC ← dd8 ²
9? 1001 xxmm dd8 JN dd8 if N=1 then PC ← dd8 ²
A? 1010 xxmm dd8 JZ dd8 if Z=1 then PC ← dd8 ²
B? 1011 xxmm dd8 JC dd8 if C=1 then PC ← dd8 ²
C? 1100 xxmm dd8 JSR dd8 MEM(dd8) ← PC

PC ← dd8 + 1 ²
D? 1101 rrxx - NEG rr rr ← 0 – rr, ou seja, rr ← ¬ rr + 1 t t t
E? 1110 rrxx - SHR rr 0→ [ registrador ] → C t t t
Fx 1111 xxxx - HLT Para a execução
rr - indica um registrador (A, B ou X);
mm - indica um modo de endereçamento (direto, indireto, imediato ou indexado);
x - indica que o bit não importa para a execução da instrução;
dd8 - representa um endereço de 8 bits;
t - indica que o código de condição é testado pela unidade de controle e ajustado de acordo;
A ausência de t indica que o código de condição não é alterado e mantém seu estado anterior;
(1) o carry gerado na instrução SUB é o inverso do borrow, ou seja, C=1 indica que não houve borrow e C=0
indica que ocorreu borrow;
(2) as instruções de desvio no modo imediato são tratadas como NOP e o segundo byte é ignorado;

Características das sub-rotinas no RAMSES
Recursividade Não permitida A capacidade de uma sub-rotina de chamar-se a si própria
Reentrância Não permitida A possibilidade de vários programas compartilharem uma única cópia da

mesma sub-rotina
Aninhamento Níveis ilimitados A possibilidade de uma sub-rotina ser chamada dentro de outra sub-rotina

Passagem de parâmetros:
Quanto ao tipo dos parâmetros

Passagem por VALOR Passagem por REFERÊNCIA
• O parâmetro é o DADO propriamente dito
• Antes da chamada copia-se o VALOR original para a
memória ou registrador
• Dentro da rotina utiliza-se a cópia da memória ou
registrador (modo direto)
• Não permite a atualização do valor original, só é
capaz de alterar o parâmetro local

• O parâmetro é o PONTEIRO para o dado
• Antes da chamada – Copia-se o ENDEREÇO do
valor original para a memória ou registrador – O
ponteiro é passado por VALOR
• Dentro da rotina – Utiliza-se o acesso aos dados
através do ponteiro – Modo INDIRETO do RAMSES
• Pode-se alterar o valor original que foi passado –
Alteração através do ponteiro
• Não permite a alteração do ponteiro original (se
existir)

LDR A vetor+3 ; Copia o valor "vetor[3]” para "param"
STR A param
JSR Inc
HLT

Inc: NOP
LDR A param ; Operação realizada sobre a cópia
ADD A #1
STR A param ; Altera apenas o parâmetro copiado
JMP Inc,I ; Não altera o valor original em "vetor[3]”

LDR A #vetor+3 ; Salva o endereço de "vetor[3]”
STR A param
JSR Inc
HLT

Inc: NOP
LDR A param,I ; Operação utilizando ponteiro
ADD A #1
STR A param,I ; Altera o valor original no “vetor[3]”
JMP Inc,I

Quanto ao local de armazenamento dos parâmetros
Passagem por REGISTRADOR Passagem em MEMÓRIA

• O parâmetro está nos registradores do processador
• Antes da chamada copia-se para um
REGISTRADOR o valor original ou ponteiro
• Dentro da rotina
– Se for um VALOR, usa-se diretamente
– Se for um PONTEIRO, depende dos modos de
endereçamento
• É muito eficiente

• Em área específica de dados
– Ver exemplos “por valor” e “por referência”
• Em área após a chamada
– A “área específica” está imediatamente após a
chamada da rotina
• É menos eficiente do que a passagem por registrador
• A rotina deve efetuar o ajuste do endereço
de retorno

LDR B vetor+3 ; Copia o "vetor[3]” para o RB
LDR X #vetor ; Salva o endereço de "vetor[0]” em RX
JSR Inc
HLT

Inc: NOP
ADD B #1 ; Operação sobre o valor copiado (RB)

LDR A #1
ADD A 3,X ; Operação utilizando ponteiro (RX)
STR A 3,X ; Altera o valor original no “vetor[3]”

JMP Inc,I

JSR Soma ; Soma os 2 bytes que seguem
DB 5 ; Parâmetro 1
DB 7 ; Parâmetro 2
HLT

Soma: NOP
LDR A Soma,I ; Pega o parâmetro "p1"

LDR B Soma ; Incrementa o ponteiro para "p2"
ADD B #1
STR B Soma ; Ajusta o endereço de retorno

ADD A Soma,I ; Soma “p1” com o parâmetro "p2"

LDR B Soma ; Incrementa o ponteiro para retornar
ADD B #1
STR B Soma ; Ajusta o endereço de retorno

JMP Soma,I

Quanto ao acesso
Acesso Absoluto Acesso Relativo

• A referência (ponteiro) indica a posição de
memória do parâmetro
• O valor a ser processado é passado em endereço
absoluto (um “parâmetro”)

• A referência (ponteiro) indica a posição do
parâmetros, a partir de uma posição de
referência
• Ver valor passado no registrador “X”

LDR A #vetor ; RA = vetor[0] + vetor[1]
STR A param
JSR SAbs

LDR A #vetor+5 ; RA = vetor[5] + vetor[6]
STR A param
JSR SAbs

SAbs: NOP
LDR A param,I ; “param” é um ponteiro pro vetor

LDR B param ; Incrementa o ponteiro
ADD B #1
STR B param

ADD A param,I ; RA = vetor[i] + vetor[i+1]
JMP SAbs,I

LDR X #0 ; RA = vetor[0] + vetor[1]
JSR SRel

LDR X #5 ; RA = vetor[5] + vetor[6]
JSR SRel
HLT

SRel: NOP
LDR A vetor,X ; Carrega em RA o vetor[x]
ADD X #1 ; Incrementa o índice X do vetor
ADD A vetor,X ; Soma o RA com o vetor[x]
JMP SRel,I

SRel2: NOP ; Alternativa sem incrementar o X
LDR A vetor,X ; Carrega em RA o vetor[x]
ADD A vetor+1,X ; Soma o RA com o vetor[x+1]
JMP SRel2,I

Tabela ASCII
H0? H1? H2? H3? H4? H5? H6? H7?
x000 x001 x010 x011 x100 x101 x110 x111

H?0 0000 null dle 0 @ P ` p
H?1 0001 soh dc1 ! 1 A Q a q
H?2 0010 stx dc2 “ 2 B R b r
H?3 0011 etx dc3 # 3 C S c s
H?4 0100 eot dc4 $ 4 D T d t
H?5 0101 enq nak % 5 E U e u
H?6 0110 ack syn & 6 F V f v
H?7 0111 bell etb ‘ 7 G W g w
H?8 1000 bsp can ( 8 H X h x
H?9 1001 ht em ) 9 I Y i y
H?A 1010 lf sub * : J Z j z
H?B 1011 vt esc + ; K [ k {
H?C 1100 ff fs , < L \ l |
H?D 1101 cr gs - = M ] m }
H?E 1110 so rs . > N ^ n ~
H?F 1111 si us / ? O _ o del

